

A MicroNET APPLICATION

William R. Lilegdon
Pritsker & Associates, Inc.
P.O. Box 2413
West Lafayette, IN 47906

Joseph J. Talavage
Professor, Purdue University
Consultant, Pritsker & Associates
West Lafayette, IN 47906

INTRODUCTION

MicroNET is a complete network simulation system designed specifically for micro-computers. Models of a wide variety of systems can be built, simulated, and analyzed using MicroNET's network approach. A network model is a graphical representation of a problem situation, that provides input data to the system and a means for describing the problem. Models are built by describing the problem situation graphically using MicroNET modeling symbols. These symbols are linked together and provide a graphic description of the problem situation. The MicroNET symbols translate directly into input statements that are recognized by the software. The MicroNET software uses a simulation procedure to analyze the network models. This procedure involves the generation of transactions, the processing of the transactions through the network, and the collection and reporting of statistical information.

The MicroNET modeler interacts with the software through the system interface. The MicroNET system interface provides a structured approach to analyzing systems using simulation. Models are built, simulated, and analyzed interactively using this interface. MicroNET divides the tasks involved in solving problems using network modeling into three subsystems: CONTROL, DEVELOPMENT, and EXECUTION. For each subsystem MicroNET provides a corresponding set of commands that perform the required tasks. To initiate any action, the user responds to a prompt from the system by entering the command for the action desired.

The CONTROL subsystem provides model management assistance. It allows models to be copied, deleted, and appended to one another. Using the CONTROL subsystem commands, the MicroNET system can be tailored to utilize additional memory and a parallel printer.

The DEVELOPMENT subsystem supports the creation and modification of MicroNET network models with full editing capabilities. Commands are also included to display the contents of any MicroNET model on the screen or printer.

Execution and analysis of MicroNET network models is performed using the EXECUTION subsystem. Commands are available to simulate a network model and display ongoing statistical information during model execution. Statistical reports may be prepared from this subsystem and presented at the screen or printer.

To describe the MicroNET approach to analyzing systems using network models we will discuss an example application. The example system is an automated inspection area. We will briefly describe the system, discuss the transition from system to model, review the model and its outputs and discuss the performance and application of the MicroNET software.

System Description

The system to be analyzed is a proposed design for the inspection area of large mechanical devices. These units arrive to the inspection area from a synchronized assembly area. The units are processed through 4 test stations. Units are moved throughout the inspection area using transfer carts. Figure 1 shows the flow of units through the area and the positioning of the transfer carts.

Each separate line of the same station is called a house. For example, the upper left most station on the figure is station 1, house 4. Note that houses of station 1 are tied directly to houses of station 2. Therefore, a unit completing service in house 1 station 1 may be blocked from continuing because the preceding unit is continuing processing in house 1 station 2.




FIGURE 1. System Schematic Automated Inspection Area

The transfer carts may remove a unit from a house and advance it to the next station but to a different house number. This alleviates the blocking potential of stations 1 and 2.

The inspection times for this system are available in tabular form. In the table each time is associated with a probability. MicroNET uses this information directly. MicroNET also supports samples from random distributions.

Below is the inspection time for station 1.

Inspection Time

First test		Second test	
%	Value	%	Value
14	80	21	50
29	85	24	55
26	90	22	60
2	95	4	65
3	105	4	75
1	180	2	150
1	300	23	30
24	60		

Inspect time station 1 = first test + second test.

Figure 1 reflects the findings of a capacity analysis previously performed. Notice that certain stations have four

houses while others have only two. The model should be created so that this capacity analysis may be evaluated. The model should also determine if the design of the inspection area provides sufficient throughput so as not to block the assembly area preceeding. As units arrive they must be moved onto the first transfer cart before the next unit arrives. If this does not occur the assembly area throughput is affected and the inspection area design must be improved. Therefore, the system should be modeled such that utilization of the stations and transfer carts be available.

It is assumed that the availability of material handling equipment to remove units from station 4 is infinite. It is also assumed that the line will be operated around the clock, in continuous shifts (1440 minutes/day).

Model Description

As previously stated, MicroNET supports the network approach to simulation modeling. The symbols will be linked together to represent the interaction of the transfer carts and inspection stations. Below the arrival of the units and the modeling of this process will be discussed in detail.

The transactions for this model are the individual units to be inspected. They will flow through symbols that represent actual time durations and control logic of the system.

The units are assembled on a synchronized assembly line before arrival to the inspection area. Because of the synchronized movement of the assembly area the time between arrivals to the inspection area is a constant, 45 minutes. The creation and arrival of units is modeled using a CREATE node. The CREATE node specifies the time between arrivals along with other operating parameters, not described here. For this model the CREATE node is:




FIGURE 2. CREATE NODE

Where the interarrival time is defined as a constant.

```
CONSTANT CYCLE_TIME 45;
```

This node will schedule the arrival of units every 45 minutes starting at time zero.

The limited capacity of the stations and transfer carts will be modeled using MicroNET resources. Resources are defined as an entity which is required by a transaction before the transaction can proceed through the network. Resources are named by the modeler to differentiate among different resources that may be required by a transaction.

The request and acquisition of a resource by a transaction begins with the arrival of a transaction at an RWAIT node. The RWAIT is where transactions request a specified number of units of a specific resource. The RWAIT node, figure 3, is used to request one unit of the resource TC1. Resource TC1 in this model is a single capacity resource representing the first transfer cart.




FIGURE 3. RWAIT Node
Transfer Cart 1

An arriving transaction requests the transfer cart, TC1. If it is available, the transaction will continue to the next node. If the transaction cannot acquire the resource, it is placed in a queue. The queue is identified at the node by name. In this example, the queue is named TC1_QUEUE. The queue is defined to hold a limited number of transactions and rank the waiting transactions according to a specified rule. The TC1_QUEUE is defined with the following statement:

```
QUEUE TC1_QUEUE 1 ;
```

The queue will have a capacity of one unit. The ranking rule in this case is of no consequence. One transaction can be stored while it waits for the first transfer cart. If another unit arrives, it will branch to another node or balk.

Balking is when a transaction cannot be stored in a queue. In terms of the inspection area it represents a unit arriving from the assembly area before the previous unit has been serviced by the first transfer cart. For this analysis this situation represents a serious design flaw and statistics are collected on the frequency of this occurrence.

The statistics are collected on the balking units using a BCOLLECT. The BCOLLECT generates the following set of statistics and does not effect transaction flow.

```
Number of observations
Average time between arrivals
Minimum value
Maximum value
Time of first
```

The three nodes discussed are combined to model the arrival of units from the assembly area and their processing by the transfer cart.




FIGURE 4. Arriving Units Submodel

The transaction, once it has acquired the first transfer cart, will be moved to the first house of station 1. The control logic of the transfer cart does this automatically because this move will be required regardless of the eventual destination of the unit. To model this in MicroNET the transaction must be delayed while the move is taking place. An ACTIVITY node, figure 6, is used to delay transactions for the specified amount of time. The time may be specified as a constant or a sample from a defined random distribution.




FIGURE 5. Activity Node
Move to 1st House

The duration is defined as 1ST_HOUSE.

After the duration of the MOVE_TO_1ST activity is completed, the transaction must select which house of station 1 to move to. From figure 1, four houses are available. The house is selected by using RWAITS with zero capacity queues.

The units arrive to RWAITS representing requests for each of the houses of station 1. Associated with each RWAIT is a zero capacity queue. When a transaction arrives it requests the associated resource. If that resource, representing a house, is free the transaction will continue to the next node. If the resource is busy or being held by another

transaction, the transaction will balk from the current RWAIT to a RWAIT representing the next house. The figure below shows the RWAITS and how they are linked together by balking. An implied priority is given to house 1, house 2, etc.

If a transaction requests each of the resources, and none are available the transaction will be placed at a GWAIT node. A GWAIT node is similar to a RWAIT node. However, at a GWAIT node the transaction waits dependent on the status of a gate and not the capacity of a resource. The function of this gate and its effect on the selection of a house will be discussed later.

After a transaction has selected a house the cart moves the unit to the appropriate house. Each of these moves is modeled by using a different activity. These represent the different times required to move the unit from its position in front of house 1 to the selected house. Note that no activity is associated with the selection of house 1. Once the cart is moved to the appropriate house, the unit is unloaded from the cart into the house.

At this point the unit is now inside station 1 and the transfer cart is released. To release a resource the FREE node is used. The FREE node, figure 7, is used to release the transfer cart so that it may service a transaction waiting from the assembly area. The transaction passes through the FREE node in zero simulated time. However, before the transaction is moved to the next node the resource freed examines the RWAITS associated with it to determine if a transaction may be moved ahead.




FIGURE 6. RWAIT Nodes Selection of House at Station 1




FIGURE 7. FREE Node
Release Transfer
Cart 1

The arrival of units to the system and their processing into the first stations is modeled in figure 8. The model of the entire system is shown in figure 9.




FIGURE 8. Arrival of Units and First Station Processing




FIGURE 9. Model of Entire System

The above nodes are translated directly into input statements that the MicroNET software will recognize. These input statements are shown below.

```

0 ( MODEL LINE_800 PAGE 5 )
1 ( NETWORK STATEMENTS )
2 CREATE ARRIVAL_CYCLE_TIME ;
3 RWAIT WAIT_FOR_CART CART_Q_1 TC1 ONE , LOST_UNITS ;
4 ACTIVITY MOVE_TO_1ST 1ST_HOUSE ;
5 RWAIT WAIT_HOUSE_1 HOUSE_Q HOUSE_1 , MOVE_INT0_1 WAIT_HOUSE_2 ;
6 RWAIT WAIT_HOUSE_2 HOUSE_Q HOUSE_2 , , WAIT_HOUSE_3 ;
7 ACTIVITY MOVE_TO_2ND 2ND_HOUSE MOVE_INT0_2 ;
8 RWAIT WAIT_HOUSE_3 HOUSE_Q HOUSE_3 , , WAIT_HOUSE_4 ;
9 ACTIVITY MOVE_TO_3RD 3RD_HOUSE MOVE_INT0_3 ;
10 RWAIT WAIT_HOUSE_4 HOUSE_Q HOUSE_4 , , NO_HOUSES ;
11 ACTIVITY MOVE_TO_4TH 4TH_HOUSE MOVE_INT0_4 ;
12 RWAIT NO_HOUSES NO_HOUSE_Q WAIT_HOUSE WAIT_HOUSE_1 ;
13 BCOLLECT LOST_UNITS ;
14 TERMINATE LEAVE ;
15

```

```

0 ( MODEL LINE_800 PAGE 6 )
1 ( SECTION 2 LINE 1 )
2 ACTIVITY MOVE_INT0_1 STA#1_LOAD ;
3 FREE REL_TC1_1 TC1 ;
4 ACTIVITY STA#1_TEST_1 STA#1_TIME ;
5 RWAIT WAIT_STA#2_1 STA#2_Q1 STA#2_1 ONE ;
6 ACTIVITY MOVE_TO_STA#2_1 STA#2_LOAD ;
7 FREE REL_HOUSE_1 HOUSE_1 ;
8 OPEN CHCK_GWA_1 WAIT_HOUSE ;
9 CLOSE CLS_GWA_1 WAIT_HOUSE ;
10 ACTIVITY STA#2_TEST_1 STA#2_TIME ;
11 RWAIT WAIT_TC2_1 TC2_Q_1 TC2 ONE ;
12 ACTIVITY MOVE&LOAD_1 MOVE&LOAD_TIME ;
13 FREE REL_STA#2_1 STA#2_1 ONE TRANSFER ;
14
15

```

```

0 ( MODEL LINE_800 PAGE 7 )
1 ( SECTION 2 LINE 2 )
2 ACTIVITY MOVE_INT0_2 STA#1_LOAD ;
3 FREE REL_TC1_2 TC1 ;
4 ACTIVITY STA#1_TEST_2 STA#1_TIME ;
5 RWAIT WAIT_STA#2_2 STA#2_Q2 STA#2_2 ONE ;
6 ACTIVITY MOVE_TO_STA#2_2 STA#2_LOAD ;
7 FREE REL_HOUSE_2 HOUSE_2 ;
8 OPEN CHCK_GWA_2 WAIT_HOUSE ;
9 CLOSE CLS_GWA_2 WAIT_HOUSE ;
10 ACTIVITY STA#2_TEST_2 STA#2_TIME ;
11 RWAIT WAIT_TC2_2 TC2_Q_2 TC2 ONE ;
12 ACTIVITY MOVE&LOAD_2 MOVE&LOAD_TIME ;
13 FREE REL_STA#2_2 STA#2_2 ONE TRANSFER ;
14
15

```

```

0 ( MODEL LINE_800 PAGE 8 )
1 ( SECTION 2 LINE 3 )
2 ACTIVITY MOVE_INT0_3 STA#1_LOAD ;
3 FREE REL_TC1_3 TC1 ;
4 ACTIVITY STA#1_TEST_3 STA#1_TIME ;
5 RWAIT WAIT_STA#2_3 STA#2_Q3 STA#2_3 ONE ;
6 ACTIVITY MOVE_TO_STA#2_3 STA#2_LOAD ;
7 FREE REL_HOUSE_3 HOUSE_3 ;
8 OPEN CHCK_GWA_3 WAIT_HOUSE ;
9 CLOSE CLS_GWA_3 WAIT_HOUSE ;
10 ACTIVITY STA#2_TEST_3 STA#2_TIME ;
11 RWAIT WAIT_TC2_3 TC2_Q_3 TC2 ONE ;
12 ACTIVITY MOVE&LOAD_3 MOVE&LOAD_TIME ;
13 FREE REL_STA#2_3 STA#2_3 ONE TRANSFER ;
14
15

```

```

0 ( MODEL LINE_800 PAGE 9 )
1 ( SECTION 2 LINE 4 )
2 ACTIVITY MOVE_INT0_4 STA#1_LOAD ;
3 FREE REL_TC1_4 TC1 ;
4 ACTIVITY STA#1_TEST_4 STA#1_TIME ;
5 RWAIT WAIT_STA#2_4 STA#2_Q4 STA#2_4 ONE ;
6 ACTIVITY MOVE_TO_STA#2_4 STA#2_LOAD ;
7 FREE REL_HOUSE_4 HOUSE_4 ;
8 OPEN CHCK_GWA_4 WAIT_HOUSE ;
9 CLOSE CLS_GWA_4 WAIT_HOUSE ;
10 ACTIVITY STA#2_TEST_4 STA#2_TIME ;
11 RWAIT WAIT_TC2_4 TC2_Q_4 TC2 ONE ;
12 ACTIVITY MOVE&LOAD_4 MOVE&LOAD_TIME ;
13 FREE REL_STA#2_4 STA#2_4 ONE TRANSFER ;
14
15

```

```

0 ( MODEL LINE_800 PAGE 10 )
1 ( SECTION 3 )
2 ACTIVITY TRANSFER ONE ;
3 RWAIT WAIT_STA#3 STATION#3_Q STATION#3 ;
4 ACTIVITY MOVE&UNLOAD TRANS_TIME2 REL_TC2 ;
5 FREE REL_TC2 TC2 ;
6 ACTIVITY STA#3_TEST STA#3_TIME ;
7 RWAIT WAIT_FOR_TC3 TC3_Q TC3 ;
8 ACTIVITY MOVE&LOAD TRANS_TIME2 ;
9 FREE REL_STA#3 STATION#3 ;
10 RWAIT WAIT_STA#4 STATION#4_Q STATION#4 ;
11 ACTIVITY MOVE&UNL_2 TRANS_TIME2 ;
12 FREE REL_TC3 TC3 ;
13 ACTIVITY STA#4_TEST CYCLE_TIME REL_STA#4 ;
14 FREE REL_STA#4 STATION#4 ;
15 TERMINATE LEAVE_SYS 1 ;

```

Notice their correspondence to the graphical symbols used above. Along with the nodes some definitions and distributions are required. They are shown below.

MicroNET Model: LINE_800

```

0  ( MODEL LINE_800 PAGE 1 )
1  ( RESOURCE DEFINITIONS )
2  RESOURCE HOUSE_1 1 , WAIT_HOUSE_1 ;
3  RESOURCE HOUSE_2 1 , WAIT_HOUSE_2 ;
4  RESOURCE HOUSE_3 1 , WAIT_HOUSE_3 ;
5  RESOURCE HOUSE_4 1 , WAIT_HOUSE_4 ;
6  RESOURCE TC1 1 , WAIT_FOR_CART ;
7  RESOURCE TC2 1 , WAIT_TC2_1 WAIT_TC2_2 WAIT_TC2_3
8                                WAIT_TC2_4 ;
9  RESOURCE TC3 1 , WAIT_FOR_TC3 ;
10 RESOURCE STA#2_1 1 , WAIT_STA#2_1 ;
11 RESOURCE STA#2_2 1 , WAIT_STA#2_2 ;
12 RESOURCE STA#2_3 1 , WAIT_STA#2_3 ;
13 RESOURCE STA#2_4 1 , WAIT_STA#2_4 ;
14 RESOURCE STATION#4 2 , WAIT_STA#4 ;
15 RESOURCE STATION#3 4 , WAIT_STA#3 ;

0  ( MODEL LINE_800 PAGE 2 )
1  ( QUEUE DEFINITIONS )
2  QUEUE CART_Q_1 0 ;
3  QUEUE HOUSE_Q_0 ;
4  QUEUE NO_HOUSE_Q_1 ;
5  QUEUE TC2_Q_1 1 ;    QUEUE STA#2_Q1 1 ;
6  QUEUE TC2_Q_2 1 ;    QUEUE STA#2_Q2 1 ;
7  QUEUE TC2_Q_3 1 ;    QUEUE STA#2_Q3 1 ;
8  QUEUE TC2_Q_4 1 ;    QUEUE STA#2_Q4 1 ;
9
10 QUEUE STATION#3_Q_1 ;
11 QUEUE TC3_Q_4 ;
12 QUEUE STATION#4_Q_1 ;
13
14
15

0  ( MODEL LINE_800 PAGE 3 )
1  GATE WAIT_HOUSE_CLS , NO_HOUSES ;
2  CONSTANT ONE 1 ;
3  CONSTANT CYCLE_TIME 45 ;
4  CONSTANT STA#1_LOAD 1 ;
5  CONSTANT STA#2_LOAD 1 ;
6  CONSTANT 1ST_HOUSE 1 ;
7  CONSTANT 3RD_HOUSE 1 ;
8  CONSTANT STA#2_TIME 65 ;
9  UNFRM 2ND_HOUSE 0 1 ;
10 UNFRM 4TH_HOUSE 1 2 ;
11 UNFRM MOVE&LOAD_TIME 0 2 ;
12 UNFRM TRANS_TIME2 0 2 ;
13
14
15

0  ( MODEL LINE_800 PAGE 4 )
1  TABLE STA#3_PART_1 20 50 24 60 1 65 25 75 3 90 3 105 24 45 ;
2  TABLE STA#3_PART_2 1 40 4 50 15 110 7 155 6 170 5 200 9 230 53
3  20 ;
4  ASSOP STA#3_TIME STA#3_PART_1 + STA#3_PART_2 ;
5  TABLE STA#1_PART_1 14 80 29 85 26 90 2 95 3 105 1 180 1 300
6  24 60 ;
7  TABLE STA#1_PART_2 21 50 24 55 22 60 4 65 4 75 2 150 23 30 ;
8  ASSOP STA#1_TIME STA#1_PART_1 + STA#1_PART_2 ;
9
10
11
12
13
14
15

```

These definitions represent activity durations, queue capacities, and resources referenced in the model.

Simulation Output and Analysis

The first step in analyzing the model of the inspection area is to validate the operation of the model. Space does not permit discussion of the validation of the entire model, however, the first events will be examined. The MicroNET Trace Report is used to validate the simulation model. The Trace Report is a textual description of the events occurring in the simulation. The Trace Report for the first four events is shown below.

remains free. The transaction continues into an activity which represents the first inspection test. The next event occurs at time 45 with the creation of the second transaction. This transaction arrives at the RWAIT node WAIT_FOR_CART, the cart is free and is seized by this transaction. This transaction then moves into the activity MOVE_TO_1ST. The MicroNET Trace Report would be evaluated in this manner for the remaining network logics to determine if the model is operating correctly.

Another output of the MicroNET system is the Summary Report. The Summary Report provides statistical information about the events occurring in the sampled time.

MicroNET Trace Report

Run Name: TEST OF INSPECT

Model Name: LINE_800

Simulation completion time is 1440
Simulation termination count is 32000

At time	0	transaction #	1 created at ARRIVAL
At time	0	transaction #	1 arrives at WAIT_FOR_CART
At time	0	transaction #	1 begins activity MOVE_TO_1ST
At time	1	transaction #	1 completed activity MOVE_TO_1ST
At time	1	transaction #	1 arrives at WAIT_HOUSE_1
At time	1	transaction #	1 begins activity MOVE_INT0_1
At time	2	transaction #	1 completed activity MOVE_INT0_1
At time	2	transaction #	1 arrives at REL_TC1_1
At time	2	transaction #	1 begins activity STA#1_TEST_1
At time	45	transaction #	2 created at ARRIVAL
At time	45	transaction #	2 arrives at WAIT_FOR_CART
At time	45	transaction #	2 begins activity MOVE_TO_1ST
At time	46	transaction #	2 completed activity MOVE_TO_1ST
At time	46	transaction #	2 arrives at WAIT_HOUSE_1
At time	46	transaction #	2 arrives at WAIT_HOUSE_2
At time	46	transaction #	2 begins activity MOVE_TO_2ND
At time	47	transaction #	2 completed activity MOVE_TO_2ND
At time	47	transaction #	2 begins activity MOVE_INT0_2

The simulation begins with the first transaction being created at the node ARRIVAL. This represents the arrival of the first unit from the assembly area. The Trace Report describes the flow of this transaction at time zero. The transaction leaves the CREATE node and arrives at the RWAIT node WAIT_FOR_CART. Obviously, the cart is available and the transaction moves to the activity node MOVE_TO_1ST. The transaction begins this activity. The activity is completed at time one and is shown as the fourth line of the trace report. The first transaction correctly chooses the first house of station 1 and begins the activity of moving into station 1. One time unit later the transaction completes this

activity and arrives to the FREE node REL_TC1_1. The transfer cart is released and polls the RWAIT nodes associated with it to find if any units have arrived that required the cart. Another unit has not arrived, therefore, the transfer cart

Below is the Summary Report for the first 24 hours of operation of the system. Each individual report will be examined and discussed.

The first report is the Activity Report. The Activity Report provides information about the delays transactions have experienced at the activities included in the model. This report provides some of the information required in the analysis of the inspection system. The utilization of the various stations may be derived from this report. The Activity Report can be used to determine the utilization of various stations. For example, the house 1 of station 1, STA#1_TEST_1, in this first day was utilized to the 80% level. However, the fourth house STA#1_TEST_1, was only 50% utilized. Obviously more observations are required, but the relevant information is presented in this report.

MicroNET
Summary ReportRun Name: TEST OF INSPECT Date: 9/1/83
Model Name: LINE_800

By: PRITSKER & ASSOC.

Beginning time of simulation 0
Ending time of simulation 1440

Statistical arrays cleared at 0

Current simulation time 1440

Activity Report

ACTIVITY LABELS	TOTAL # OF COMPLETIONS	SERVER UTIL. FRACTION	AVERAGE TIME IN ACTIVITY	MINIMUM # IN ACTIVITY	MAXIMUM # IN ACTIVITY	CURRENT # IN ACTIVITY
MOVE_TO_1ST	32	0.0	1.0	0	1	1
MOVE_TO_2ND	9	0.0	0.4	0	1	0
MOVE_TO_3RD	8	0.0	1.0	0	1	0
MOVE_TO_4TH	6	0.0	1.1	0	1	0
MOVE_INTD_1	9	0.0	1.0	0	1	0
STAN1_TEST_1	8	0.8	127.5	0	1	1
MOVE_TO_STAN2_1	8	0.0	1.0	0	1	0
STAN2_TEST_1	8	0.3	65.0	0	1	0
MOVE&LOAD_1	8	0.0	0.7	0	1	0
MOVE_INTD_2	9	0.0	1.0	0	1	0
STAN1_TEST_2	8	0.7	128.7	0	1	1
MOVE_TO_STAN2_2	8	0.0	1.0	0	1	0
STAN2_TEST_2	7	0.3	65.0	0	1	1
MOVE&LOAD_2	7	0.0	1.2	0	1	0
MOVE_INTD_3	8	0.0	1.0	0	1	0
STAN1_TEST_3	7	0.7	134.2	0	1	1
MOVE_TO_STAN2_3	7	0.0	1.0	0	1	0
STAN2_TEST_3	7	0.3	65.0	0	1	0
MOVE&LOAD_3	7	0.0	1.2	0	1	0
MOVE_INTD_4	6	0.0	1.0	0	1	0
STAN1_TEST_4	6	0.5	135.8	0	1	0
MOVE_TO_STAN2_4	6	0.0	1.0	0	1	0
STAN2_TEST_4	6	0.2	65.0	0	1	0
MOVE&LOAD_4	6	0.0	1.3	0	1	0
TRANSFER	28	0.0	1.0	0	1	0
MOVE&UNLOAD	28	0.0	0.7	0	1	0
STAN3_TEST	26	2.4	127.8	0	4	2
MOVE&LOAD	26	0.0	0.9	0	1	0
MOVE&UNL_2	26	0.0	1.0	0	1	0
STAN4_TEST	25	0.7	45.0	0	2	1

Queue Report

QUEUE LABELS	TOTAL # OF NODE DEPARTURES	# OF ZERO WAITS	AVERAGE WAITING TIME	AVERAGE # IN QUEUE	MINIMUM # IN QUEUE	MAXIMUM # IN QUEUE	CURRENT # IN QUEUE
CART_Q_1	33	33	n.a.	0.0	0	0	0
HOUSE_Q	32	32	n.a.	0.0	0	0	0
NO_HOUSE_Q	0	0	n.a.	0.0	0	0	0
TC2_Q_1	8	8	n.a.	0.0	0	0	0
STAN2_Q1	8	8	n.a.	0.0	0	0	0
TC2_Q_2	7	6	2.0	0.0	0	1	0
STAN2_Q2	8	8	n.a.	0.0	0	0	0
TC2_Q_3	7	7	n.a.	0.0	0	0	0
STAN2_Q3	7	7	n.a.	0.0	0	0	0
TC2_Q_4	6	5	2.0	0.0	0	1	0
STAN2_Q4	6	6	n.a.	0.0	0	0	0
STATION#3_Q	28	25	9.0	0.0	0	1	0
TC3_Q	26	26	n.a.	0.0	0	0	0
STATION#4_Q	26	24	9.0	0.0	0	1	0

Resource Report

RESOURCE LABELS	TOTAL # OF RESRC. CAPTURES	AVERAGE IN USE	MINIMUM IN USE	MAXIMUM IN USE	CURRENT IN USE	TOTAL TIME IDLE ALL UNITS
HOUSE_1	9	0.8	0	1	1	270
HOUSE_2	9	0.7	0	1	1	346
HOUSE_3	8	0.7	0	1	1	390
HOUSE_4	6	0.5	0	1	0	606
TC1	33	0.0	0	1	1	1357
TC2	28	0.0	0	1	0	1332
TC3	26	0.0	0	1	0	1371
STAN2_1	8	0.3	0	1	0	906
STAN2_2	8	0.3	0	1	1	904
STAN2_3	7	0.3	0	1	0	969
STAN2_4	6	0.2	0	1	0	1034
STATION#4	26	0.8	0	2	1	592
STATION#3	28	2.4	0	4	2	205

Time in System Report

TERM NODE LABELS	TOTAL # OF NODE DEPARTURES	AVERAGE TIME IN SYSTEM	MINIMUM TIME IN SYSTEM	MAXIMUM TIME IN SYSTEM
LEAVE	0	n.a.	n.a.	0
LEAVE_SYS	25	383.6	291	524

COLLECT STATISTIC REPORT

COLLECT LABELS	# OF OBSERVATIONS	AVERAGE VALUE	MINIMUM VALUE	MAXIMUM VALUE	TIME OF FIRST
LOST_UNITS	n.a.	0.0	n.a.	0	0

The Queue Report provides information about blocking that has occurred at the various stations. The cart queue that represents arriving units from assembly is WAIT_FOR CART. The first column of the report describes the number of transactions passing through the queue. The second column contains the number of transactions that arrived to the queue and were immediately serviced. For this case both numbers are equal. This means that every unit arriving from the assembly area was processed immediately. This was also a decision criterion of our design and evaluation.

The queue associated with the transactions waiting for transfer cart 2 from house 2 of station 2 is labeled WAIT_TC2_2. This information reveals that 6 out of 7 transactions were immediately processed and the one transaction that waited, waited for 2 minutes. The report also reflects, from the last column, that the number in this queue is zero.

The Resource Report reflects information about the utilization of the stations. This information should corroborate the information derived from the Activity Report. Of special interest is the utilization of the three transfer carts. From the Resource Report the exact utilization of the transfer cart may be derived. The right column of this report reflects the total time idle by the 3 single capacity transfer cart resources. To derive the utilization of the transfer cart, simply subtract the value in this column from the total simulation time and then divide by the total simulation time. For transfer cart 1 the utilization is 5.7%. This reflects that perhaps a different method of material handling should be examined to reduce the cost of the design and gain a better utilization of the material handling equipment.

The Time in System Report reflects information about the throughput of units through the inspection area. The TERMINATE node LEAVE_SYS has experienced 25 departures in this time period. These 25 units had an average time in the system of 383 minutes. This values could be used in a planning function to determine scheduling of units through this design.

The analysis of this inspection area would require further simulation of this model to gain a larger quantity of observations. However, all of the information required for complete evaluation of the design is available through the MicroNET statistical reports. The performance of the simulation software with this model has been evaluated on the IBM PC. The memory required to store and execute this model is 49K. To execute the model with the

monitor feature presenting the textual description to the screen for the 1 day/24 hour analysis 5 minutes were required. If the monitor output is suspended, the simulation is completed in 1 1/2 minutes.

Conclusion

MicroNET provides powerful network simulation capabilities on available microcomputers. This model and discussion demonstrates a subset of the capabilities that are available in the MicroNET software. MicroNET provides an alternative source of modeling power to systems analysts and decision makers utilizing an easy-to-use system interface. Continued research and development is being performed to advance the capabilities of simulation on microcomputers.